

Une enquête sur un mystère...

J'invite le lecteur à consulter le site pour des informations complémentaires.

Page d'accueil du site Internet :

[page d'accueil](#)

d'autres pdf, sur différents sujets :

[liste des PDF](#)

Dans certains montages électroniques câblés sur plaque Labdec, on subit des oscillations non désirées qui rendent la manipulation délicate. J'ai eu l'idée de montrer, par la simulation Pspice, ce phénomène, rencontré souvent en salle de Travaux Pratiques.

Bien entendu, il ne s'agit pas de simuler le circuit câblé proprement dit, car bien trop difficile à modéliser. Même si on ajoute des capacités parasites un peu partout, des résistances de contact ici et là, des inductances modélisant des fils de câblage, cela donne un schéma énorme et... sans résultat convaincant.

J'ai modélisé simplement 4 défauts liés à l'environnement, pour montrer leur contribution à l'instabilité d'un montage amplificateur. C'est un peu tiré par les cheveux, mais cela a le mérite d'exister. Une pointe d'humour est donc en cohérence avec cette étude scientifique...


La scène du délit : un TP tout simple...

Le montage classique de chez classique : une paire dif polarisée par un miroir. Et pour simplifier, on impose une entrée à la masse, on place un GBF sur l'autre, un oscillo, et là... « *monsieur, ça ne marche pas !* ».

Et l'enseignant de venir pour constater que, malgré qu'il n'y ait aucune erreur, on voit tout et n'importe quoi à l'écran...

Bref, cela oscille comme ce n'est pas permis et cela étouffe le signal...

C'est un homicide !



Menons l'enquête :

1) à l'origine de l'oscillation, il y a un couplage entre l'entrée et la sortie

Il est difficile de localiser la cause de l'oscillation. Le montage n'est pas bouclé. Même si on cherche une contre réaction provoquée par une résistance d'émetteur, cela ne forme pas un système bouclé instable.

D'où une recherche orientée par une réaction entre la sortie et l'entrée positive, ce qui conduit à un fonctionnement divergeant du système. Néanmoins, cela n'explique pas non plus un régime oscillatoire entretenu.


Pour avoir des auto-oscillations, il faut aussi d'autres couplages.

Des éléments pouvant provoquer les couplages sont des condensateurs parasites, présents sur les contacts de la plaque « Labdec ». Par une double mesure à l'aide du pont LCR400 de TTI, on a :


- entre 2 contacts voisins : ≈ 2 pF,
- entre 2 contacts en face à face : $\approx 0,4$ pF,
- entre 2 contacts séparés un contact : ≈ 1 pF.

La photo montre 2,719 pF entre 2 contacts voisins. En retirant 0,6 pF environ de condensateur résiduel (fils qui ne touchent pas les contacts), cela donne $\approx 2,1$ pF.

Ces essais ont été refaits sur une autre plaque, équipée d'une plaque métallique (non reliée) : les résultats sont identiques. La plaque de fixation sous la Labdec n'est pas en cause.



La disposition des transistors utilisés au sein du circuit intégré n'est pas régulière. Il s'ensuit que les condensateurs parasites de la plaque Labdec ne sont pas distribués de façon similaire pour les transistors, car dépendant du choix fixé lors du câblage. Cela complique un schéma de simulation qui modéliserait le montage câblé sur plaque Labdec.




2) Un coupable idéal : un condensateur parasite...

La suite repose sur l'hypothèse que la combinaison de ces différents condensateurs est équivalente à un seul, placé entre la sortie et l'entrée non inverseuse.

Pour les simulations qui vont suivre, ce condensateur a une capacité de 0,85 pF. Cette valeur est cohérente avec les condensateurs parasites mis en jeu sur la plaque Labdec.

Néanmoins, comme signalé précédemment, la présence de ce seul composant ne justifie pas une auto oscillation. Il faut chercher d'autres perturbations.



3) Son complice : une alimentation imparfaite


Il a été mesuré, à l'aide du pont LCR400 de TTI, une inductance de 0,7 μH pour un câble de 30 cm de long.

Notons que la valeur de l'inductance d'un simple bout de fil n'est pas intrinsèque au fil. Il faut tenir compte de la proximité de la masse, de la terre avoisinante...

On propose de modéliser l'alimentation utilisée sur les tables de T.P. par des sources de tension parfaites, en série avec une inductance sur le +12 V, -12 V et la masse.

On néglige l'impédance interne de la source continue.

La valeur choisie pour l'inductance des cordons est de 1,2 μH . Cette valeur reste cohérente. Pour simplifier, on néglige les mutuelles inductances...




4) Le grain de sable : le point de masse de la voie 1 de l'oscilloscope n'est pas parfait.

Il est usuel de placer un Té entre le G.B.F. et la voie 1 de l'oscilloscope. Le signal est alors injecté sur l'entrée du montage, par un cordon BNC, et la masse est alors connectée... quelque part.

Pour modéliser une « mauvaise masse », il a été inséré une liaison au GND par une inductance L3. On conserve la valeur de 1,2 μH .

Ce schéma montre également la résistance interne de 50 Ω du G.B.F., et le condensateur de 120 pF que présentent l'oscilloscope et son câble.




5) Le facteur humain : le point de masse de la voie 2 de l'oscilloscope n'est pas parfait.

Dans le TP on observe la sortie inverseuse (on verra plus loin la conséquence de décaler l'oscilloscope sur l'autre sortie).

On utilise un câble coaxial, dont la masse (si on ne l'oublie pas non plus...) est alors connectée... quelque part ailleurs....

Pour modéliser une « mauvaise masse », il a été placé une liaison au GND par une inductance L2. On conserve la valeur de 1,2 μH .



En conclusion, le schéma simulé modélisant la scène du délit est celui-ci :




Schéma simulé.

*Les n° (de 0 à 8) sont les n° de nœud de Pspice.
Un sous-circuit délivre VPLUS, GND, VMOINS.*

En rouge, les protagonistes du malheur :

- inductance des câbles de l'alimentation
- inductance des masses des coax de l'oscilloscope
- un condensateur équivalent entre l'entrée et la sortie non inverseuse

En vert, les redresseurs de torts, donnés par l'enseignant (les redresseurs, pas les torts...) :

- les condensateurs de découplage (2 x 100 nF),
- le condensateur de compensation (1 pF),
- un condensateur qui découple l'entrée (1 pF).

Pour les besoins de la reconstitution de la scène du délit, notez les subtilités :

0 (la référence Pspice),
GND (le point froid de la carte Labdec),
GND2,
GND3.

Par simplicité, ce schéma n'exploite pas :

- les boucles de masse (par exemple, les 2 voies de l'oscilloscope ont un point commun, ce qui ferait la liaison entre GND2 et GND3),
- la résistance interne de 1 MΩ des voies V1 et V2 de l'oscilloscope,
- les imperfections des composants passifs, etc.

La sortie inverseuse est V(2). On peut également observer la sortie non inversée en V(3).

Le fichier dif_osci.cir ci-dessous montre tous ces éléments.

En commentant / décommentant les lignes, on retire / place les composants.

```

00 Amplificateur différentiel oscillant
01 *
02 * fichier dif_osci.cir
03 *
04 *
05 .MODEL CA3086 NPN ( IS=10.000E-15 BF=156.66 VAF=100 IKF=36.783E-3
06 + ISE=114.82E-15 NE=1.4743 BR=.1001 VAR=100 IKR=10.01E-3
07 + ISC=10E-15 RC=10 CJE=1.026E-12 MJE=.33333 CJC=992E-15
08 + MJC=.33333 TF=278.55E-12 XTF=91.9 VTF=18.9 ITF=.77631 TR=10E-9 )
09
10 .SUBCKT ALIM VPLUS VMOINS GND
11 VP Vsup 0 12 ; alimentation + parfaite
12 Lp Vsup VPLUS 1.2uH ; inductance du fil
13 VM 0 Vinf 12 ; alimentation - parfaite
14 Lm Vinf VMOINS 1.2uH ; inductance du fil
15 Lgnd GND 0 1.2uH ; inductance du fil de masse
16 .ENDS
17
18 * circuit : entrée GBF 1, sortie 2, point froid GND
19
20 XALIM 6 7 GND ALIM
21
22 RC1 6 2 10k ; charge
23 RC2 6 3 10k ; passive.
24 Q1 2 1 4 CA3086; paire
25 Q2 3 GND 4 CA3086; différentielle.
26 Q3 4 5 7 CA3086; duplique Io
27 Q4 5 5 7 CA3086; parcouru par Io
28 RP GND 5 10k ; avec 22k n'oscille pas
29
30 * reaction positive
31 Cparasite 3 1 0.85p ; LE FAUTIF !
32
33 * stabilisation
34 *Ccompensation 1 2 1p ; classique
35
36 * découplage
37 *Cd1 6 GND 100n ; valeurs
38 *Cd2 7 GND 100n ; usuelles
39
40 * signal, entre V(1) et GND3 :
41 Vin 8 GND3 ac 1 sin (0 30m 10k 0 0 0) ;
42 L3 GND3 GND 1.2uH ; point de masse du GBF
43 Rin 8 1 50 ; interne au GBF
44
45 * Oscilloscope entre V(1) et GND3, en // par le Té :
46 Coscillo_V1 1 GND3 120pF ;
47
48 * découplage de l'entrée
49 *Cadd 8 GND 1p
50
51 * Oscilloscope entre V(2) et GND2 :
52 Coscillo_V2 2 GND2 120pf ; scope branché sortie
53 L2 GND2 GND 1.2uH ; point de masse de l'oscillo
54
55 * analyse temporelle :
56 .TRAN 1u 300u 0 1u ;
57
58 .probe
59 .end


```

Netlist

L'ENQUETE DONNE SES RÉSULTATS (DE SIMULATION)


On reproduit le « monsieur, ça ne marche pas ! »...

Le run a été interrompu après quelques μs .
 On visualise $V(2)$, c'est-à-dire \bar{S} , la sortie inversée.
 L'oscillation fait 6 V d'amplitude.



Le même, décrit très finement, par un zoom :

Une mesure donne une période de 9 ns, soit une fréquence de 111 MHz. Cette fréquence est de l'ordre de grandeur $\frac{1}{2\pi} \sqrt{\frac{1}{LC}} = 157 \text{ MHz}$, où $L = 1,2 \mu H$ et $C = 0,85 \text{ pF}$.



Vérifions que les conseils donnés en salle de TP sont efficaces...

Il a été lancé 4 runs indépendants : à chaque run, une seule intervention est mise en œuvre, les 3 autres restent désactivées.

1) Condensateur de compensation.

C'est la « parade » classique.
 Ici, 1 pF suffit.

```
33 * stabilisation
34 Ccompensation 1 2 1p ; classique
35
```

La ligne 34 est décommentée

2) Découplage des alimentations.

Pour pallier l'impédance non nulle des alimentations $\pm 12 \text{ V}$.

```
36 * découplage
37 Cd1 6 GND 100n ; valeurs
38 Cd2 7 GND 100n ; usuelles
```

Les 2 lignes 37 et 38 sont décommentées

3) Découplage de l'entrée.

Montre d'ailleurs le rôle néfaste de $L3$: sans $L3$, l'entrée serait découplée par 120 pF de la voie 1, et $C_{add} = 1 \text{ pF}$ serait inutile.

```
48 * découplage de l'entrée
49 Cadd 8 GND 1p
```

La ligne 49 est décommentée

4) Déplacement de l'oscilloscope

En chargeant ainsi la sortie S , au lieu de \bar{S} , on réduit l'influence de $C_{parasite}$.

```
51 * Oscilloscope entre V(2)
52 Coscillo_V2 3 GND2 120pf
53 L2 GND2 GND 1.2uH ;
```

En ligne 52, on place 3 à la place de 2

Verdict : il n'oscille plus, il amplifie...

Grâce à 1) **ou** 2) **ou** 3) **ou** 4), on a le montage qui fonctionne :

Portée par une composante continue de 6 V, on a en sortie $V(2)$ une sinusoïde de 3 V d'amplitude, 10 kHz, pour 30 mV injecté : on retrouve $A_d = - 100$.



Rappelons que $A_d = - R_C g_m / 2$, avec $g_m = I_C / V_T$, où $V_T = kT/q = 26 \text{ mV}$.

Le miroir fournit $I_C = (12 - 0,6) / 10k = 1,14 \text{ mA}$, d'où $I_C = 0,57 \text{ mA}$ et $g_m = 22 \text{ mA/V}$ d'où $A_d = - 109$.

Remarquons que l'on retrouve en simulation un phénomène repéré en salle de T.P. En modifiant l'amplification par un changement de R_p , on peut observer que le montage n'oscille pas : avec $R_p = 22 \text{ k}\Omega$ (au lieu de $10 \text{ k}\Omega$), la sinusoïde est portée par $9,3 \text{ V}$ et présente une amplitude de $1,4 \text{ V}$, soit une amplification différentielle dissymétrique de -47 . (Il faut alors $C_{\text{parasite}} = 5,6 \text{ pF}$ pour oscillation). Mais avait-on besoin de Pspice pour aboutir à cette remarque ?

De même, en poussant plus loin les investigations, **on peut retrouver également que l'accrochage est lié également au niveau de tension.**


Pour le vérifier, modifions légèrement le schéma :

a) on modifie la cause de l'oscillation (2 condensateurs de 2 pF au lieu de $0,85 \text{ pF}$),

b) on retire la voie 2 de l'oscilloscope.

Premier cas simulé : le signal d'entrée est de 15 mV d'amplitude, 3 kHz .

Dans ces conditions, l'amplificateur différentiel oscille, et le signal de sortie est la contribution de l'amplification de l'entrée à 15 mV + l'oscillation (mesurée par un zoom : $11,6 \text{ MHz}$).



```

XALIM 6 7 GND ALIM
RC1 6 2 10k ; charge
RC2 6 3 10k ; passive.
Q1 2 1 4 CA3086; paire
Q2 3 GND 4 CA3086; différentielle.
Q3 4 5 7 CA3086; duplique Io
Q4 5 5 7 CA3086; parcouru par Io
RP GND 5 10k ; avec 22k n'oscille pas

* reaction positive
*Cparasite 3 1 0.85p ; LE FAUTIF !
Cparasite1 3 1 2p ; ce qui fait
Cparasite2 2 GND 2p ; osciller

* stabilisation
*Ccompensation 1 2 1p ; classique

* découplage
*Cd1 6 GND 100n ; valeurs
*Cd2 7 GND 100n ; usuelles

* signal. entre V(1) et GND3 :
Vin 8 GND3 ac 1 sin (0 15m 3k 0 0 0) ; ou 25m
L3 GND3 GND 1.2uH ; point de masse du GBF
Rin 8 1 50 ; interne au GBF

* Oscilloscope entre V(1) et GND3, en // par le Té
Coscillo_V1 1 GND 120pF ;

* découplage de l'entrée
*Cadd 8 GND 1p

* Oscilloscope entre V(2) et GND2 :
*Coscillo_V2 2 GND2 120pf ; scope branché sortie
*L2 GND2 GND 1.2uH ; point de masse de l'oscillo

* analyse temporelle :
.TRAN 20n 1m 0 20n ;

probe
end


```

Nouvelle netlist

Deuxième cas simulé : le signal d'entrée est de 25 mV d'amplitude, 3 kHz .

Dans ces conditions, l'amplificateur différentiel oscille (toujours à $11,6 \text{ MHz}$) de façon « aléatoire », en fait déclenché par le niveau de tension, ce qui ajoute une non-linéarité au phénomène. Cela donne un aspect interrompu au signal de sortie.

On reconnaît un phénomène que l'on voit également en salle de T.P.



CONCLUSION : les coupables sournoisement dissimulés ont été repérés...

Rappelons que le schéma simulé n'est pas - et ne peut pas être - le circuit physiquement câblé. Néanmoins, ces runs sont proches des comportements observés en pratique. **Il est donc exact de dire qu'à cause de condensateurs inférieurs à 1 pF , ou d'inductances de l'ordre du μH , « judicieusement mal placés », l'amplificateur différentiel oscille.**

De plus, les nombreux runs que j'ai menés montrent que l'oscillation peut avoir lieu en modifiant de très très peu un condensateur, où une inductance. Cette très faible robustesse explique aussi pourquoi une simple perturbation sur l'environnement (on approche la main de la plaquette) peut dégrader, ou améliorer les choses, ce qui a le don de surprendre les étudiants, et encore parfois les enseignants...

<p>Amplificateur différentiel oscillant * * fichier dif_osci.cir * .MODEL CA3086 NPN (IS=10.000E-15 BF=156.66 VAF=100 IKF=36.783E-3 + ISE=114.82E-15 NE=1.4743 BR=.1001 VAR=100 IKR=10.01E-3 + ISC=10E-15 RC=10 CJE=1.026E-12 MJE=.33333 CJC=992E-15 + MJC=.33333 TF=278.55E-12 XTF=91.9 VTF=18.9 ITF=.77631 TR=10E-9)</p>	
<p>.SUBCKT ALIM VPLUS VMOINS GND VP Vsup 0 12 ; alimentation + parfaite Lp Vsup VPLUS 1.2uH ; inductance du fil VM 0 Vinf 12 ; alimentation - parfaite Lm Vinf VMOINS 1.2uH ; inductance du fil Lgnd GND 0 1.2uH ; inductance du fil de masse .ENDS</p> <p>.subckt INV MF in out Valim Vdd 0 DC=3.3 Mp out in vdd vdd MODP W=6u L=0.35u ; W/L grand Mn out in 0 0 MODN W=0.35u L=4u ; W/L petit .ends</p> <p>* circuit : entrée GBF 1, sortie 2, point froid GND</p> <p>XALIM 6 7 GND ALIM</p> <p>RC1 6 2 10k ; charge RC2 6 3 10k ; passive. Q1 2 1 4 CA3086; paire Q2 3 GND 4 CA3086; différentielle. Q3 4 5 7 CA3086; duplique Io Q4 5 5 7 CA3086; parcouru par Io RP GND 5 10k ; avec 22k n'oscille pas</p> <p>* réaction positive *Cparasite 3 1 0.85p ; LE FAUTIF ! Cparasite1 3 1 2p ; ce qui fait Cparasite2 2 GND 2p ; osciller</p>	<p>* stabilisation *Ccompensation 1 2 1p ; classique</p> <p>* découplage *Cd1 6 GND 100n ; valeurs *Cd2 7 GND 100n ; usuelles</p> <p>* signal, entre V(1) et GND3 : Vin 8 GND3 ac 1 sin (0 25m 3k 0 0 0) ; ou 25m L3 GND3 GND 1.2uH ; point de masse du GBF Rin 8 1 50 ; interne au GBF</p> <p>* Oscilloscope entre V(1) et GND3, en // par le Té : Coscillo_V1 1 GND 120pF ;</p> <p>* découplage de l'entrée *Cadd 8 GND 1p</p> <p>* Oscilloscope entre V(2) et GND2 : *Coscillo_V2 2 GND2 120pf ; scope branché sortie *L2 GND2 GND 1.2uH ; point de masse de l'oscillo</p> <p>* analyse temporelle : .TRAN 20n 1m 0 20n ;</p> <p>.probe .end</p>

Netlist, prête à simuler

articles 1 à 43 : sur le livre

Tableau récapitulatif des articles PDF disponibles sur ce site

n°	titre	lien présentation	lien direct article
	Guide d'installation et d'emploi simplifié	présentation	document PDF
44	Exemples basiques et des exercices...	présentation	document PDF
45	Un exemple de circuit passif	présentation	document PDF
46	Un oscillateur Colpitts	présentation	document PDF
47	Compensation en fréquence des amplificateurs opérationnels	présentation	document PDF
48	Un amplificateur à transistors bipolaires	présentation	document PDF
49	Une bascule D Flip Flop CMOS	présentation	document PDF
50	Une porte XOR à transistors MOS	présentation	document PDF
51	Un VCO à 12 transistors MOS	présentation	document PDF
52	Une PLL à moins de 20 transistors MOS	présentation	document PDF
53	Un oscillateur à résistance négative	présentation	document PDF
54	Une charge électronique	présentation	document PDF
55	Un amplificateur en classe C	présentation	document PDF
56	Le monostable 74 123	présentation	document PDF
57	Un amplificateur en classe D	présentation	document PDF
58	Le transformateur en linéaire	présentation	document PDF
59	La loi d'ohm thermique	présentation	document PDF
60	Le transformateur en non linéaire	présentation	document PDF
61	Robustesse d'un oscillateur en anneau	présentation	document PDF
62	Une alimentation stabilisée	présentation	document PDF
63	Modélisation d'un haut-parleur	présentation	document PDF
64	Un synthétiseur de fréquence	présentation	document PDF
65	Un ampli audio de Sparkfun	présentation	document PDF
66	Simulation logique et analogique	présentation	document PDF
67	Un oscillateur à relaxation	présentation	document PDF
68	Lecteur de TAG RFID 125 kHz	présentation	document PDF
69	Diagramme de l'œil avec Pspice	présentation	document PDF
70	Un amplificateur hautes fréquences	présentation	document PDF
71	Une bizarrerie enfin expliquée...	présentation	document PDF
72	Comprendre le paramétrage de la FFT	présentation	document PDF
73	La relation de Bennett	présentation	document PDF
74	Simuler un circuit à plus de 20 transistors avec PSpice Eval	présentation	document PDF
75	Une horloge biphase sans recouvrement	présentation	document PDF
76	Quelques simulations sur la diode	présentation	document PDF
77	Un ampli classe A, avec transformateur de sortie	présentation	document PDF
78	Des stimuli pour PSpice	présentation	document PDF
79	Simuler le TL431 : zener ajustable	présentation	document PDF
80	Un ADC flash	présentation	document PDF
81	Une chaîne d'acquisition : S&H, ADC, DAC	présentation	document PDF
82	Un amplificateur 50 MHz	présentation	document PDF
83	Un dérivateur non inverseur	présentation	document PDF
84	Un amplificateur bipolaire avec push pull CMOS	présentation	document PDF
85	Rôle des répéteurs logiques dans un circuit intégré	présentation	document PDF
86	Un driver logique CMOS pour charge 50 ohms	présentation	document PDF
87	Des triggers de Schmitt et des applications	présentation	document PDF
88	Un filtre gaussien analogique	présentation	document PDF
89	Un générateur de bruit rose	présentation	document PDF

90	Un anémomètre à fil chaud : simulation comportementale	présentation	document PDF
91	Un oscillateur à pont de Wien stabilisé par CTN	présentation	document PDF
92	L'emballage thermique d'une diode	présentation	document PDF
93	Les puissances dans un amplificateur	présentation	document PDF
94	Asservissement de puissance dans une résistance	présentation	document PDF
95	Asservissement de la puissance émise par une antenne radio	présentation	document PDF
96	Un driver de LED de puissance	présentation	document PDF
97	Exploiter Pspice pour simuler des filtres numériques	présentation	document PDF
98	Un filtre en cosinus surélevé avec Pspice	présentation	document PDF
99	Effet de la température sur un amplificateur en classe A	présentation	document PDF
100	Un amplificateur à transistors JFET et bipolaires	présentation	document PDF
Supplément, hors article :			
mon cours « Electronique pour les communications numériques », polycopié couleur 201 pages en pdf			

[retour à l'écran d'accueil de ce site](#)